
==============AADC_TO_ADC================

Start with a 8Bit ADDC.

Sample and hold the 8Bit AADC and with a 8msec clock, then the normal spectrum for a ADC results. 

Now this is a truly random noise floor set by the 8 bit resolution. 
And now mirror images of the input signal will appear at multiples of the sample rate. 

========Spice_Code================================
Simple_Asynchronous_ADC_clocked
*
*       VIN
*      ________________    __________    __________    __________    __________    __________    __________    
*    _|_     | TH0    _|_  | TH1    _|_  | TH0    _|_  | TH1    _|_  | TH0    _|_  | TH1    _|_  | TH0    _|_    TH1    
*   /_  \    |      _/   \ |      _/   \ |      _/   \ |      _/   \ |      _/   \ |      _/   \ |      _/   \       
*  // \  \   | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/   |\  
*  \   \//   |_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |_____| \_
*   \___/      | / | OFF0    | / |  OFF1   | / | OFF2    | / |  OFF3   | / | OFF4    | / |  OFF5   | / | OFF6    | / |  
*    _|_       |/  |         |/  |         |/  |         |/  |         |/  |         |/  |         |/  |         |/  |
*    ///          _|_           _|_           _|_           _|_           _|_           _|_           _|_           _|_
*                |D0 |         |D1 |         |D2 |         |D3 |         |D4 |         |D5 |         |D6 |         |D7 |
*                |___|         |___|         |___|         |___|         |___|         |___|         |___|         |___|
*
*        ________



*        __|     \
*        __|      \
*        __| DAC   \__
*        __|       /
*        __|      /
*        __|_____/
*
*=========Create_Signal==================
*V_SIN#   NODE_P NODE_N DC     VALUE  SIN(   V_DC   AC_MAG FREQ   DELAY  FDamp)
VIN       VIN    0      DC     0      SIN(   .5      .499     3                  )
*V_PULSE# NODE_P NODE_N DC     VALUE  PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VCLKL     CLK    0      DC     0      PULSE( 0      1      1n     1n     1n     4m     8m     )

XPOSE1    CLK    CNTL   POS_E 
XS_H1     VOUTA  CNTL   VOUT      SH 

BTH0      D0     0      V =    u( V(VIN)  -1/2)
BOFF0     VIN    OFF0   V =    V(D0)/2
BTH1      D1     0      V =    u( V(OFF0) -1/4)
BOFF1     OFF0   OFF1   V =    V(D1)/4
BTH2      D2     0      V =    u( V(OFF1) -1/8)
BOFF2     OFF1   OFF2   V =    V(D2)/8
BTH3      D3     0      V =    u( V(OFF2) -1/16)
BOFF3     OFF2   OFF3   V =    V(D3)/16
BTH4      D4     0      V =    u( V(OFF3) -1/32)
BOFF4     OFF3   OFF4   V =    V(D4)/32
BTH5      D5     0      V =    u( V(OFF4) -1/64)
BOFF5     OFF4   OFF5   V =    V(D5)/64
BTH6      D6     0      V =    u( V(OFF5) -1/128)
BOFF6     OFF5   OFF6   V =    V(D6)/128
BTH7      D7     0      V =    u( V(OFF6) -1/256)
BOFF7     OFF6   OFF7   V =    V(D7)/254
BTH8      D8     0      V =    u( V(OFF6) -1/512)
BOFF8     OFF7   OFF8   V =    V(D8)/245
BDAC      VOUTA   0      V =    V(D0)/2+V(D1)/4+V(D2)/8+V(D3)/16 +V(D4)/32 +V(D5)/64 +V(D6)/128 +V(D7)/256

.control
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      .0125m    1      0      .0125m   
set       pensize = 2
plot      vin  vout
plot      vin -vout xlimit 0 .35

echo             "===================FFT_and_Plot====================================="
linearize 
let              FFT_BandWidth_Hz =  20k
let              FFT_resolution_Hz = 1
echo             "FFT_BandWidth_Hz=  $&FFT_BandWidth_Hz"
echo             "FFT_resolution_Hz= $&FFT_resolution_Hz"
set              specwindow =         "rectangular"
spec             $&FFT_resolution_Hz  $&FFT_BandWidth_Hz   $&FFT_resolution_Hz     v(vout)
*let              freqL = length(frequency)
let              expect = (.5/256)/sqrt(128)
plot             mag (vout) expect loglog
*plot            mag (vout)  ylog xlimit 400 600
echo             "===================Done============================================="

.endc

*=========Sample_Hold============================================
*         
*            IN2                                  
*    ___  |\    /         IN3 |\      ___
*   |IN |_|1\__/  ____________|2\____|OUT|
*   |___| | / S1   _|_        | /    |___|
*         |/       ___        |/           
*                C1 |               
*                  _|_                
*                  /// 
.SUBCKT   SH     IN     CNTL      OUT   
 B1       IN2    0      V =       v(IN )
 S1       IN2    IN3    CNTL     0    SW 
 C1       IN3    0      .1u

 R1       IN3    0      100Meg
 B2       OUT    0      V =       v(IN3 )
.ENDS     SH

*=========POS_Edge============================================
*
*  ___   |\             VBF ____                ___
* |IN |__|3\________________|   \   ___    ___  |  \   ___
* |___|  | / |           VLP| PE \_|OUT|  |IN |_|PE \_|OUT|
*        |/  |/\  /\  /\____|    / |___|  |___| |   / |___|
*               \/  \/   |  |___/               |__/
*               RLP     _|_
*                       ___
*                   CLP  |
*                       _|_
*                       ///
*
.SUBCKT   POS_E  IN     OUT   
 BBUF     VBF    0      V =   u( v(IN )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      1n   IC=0
 BAND     OUT    0      V =    u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
.ENDS     POS_E

.MODEL    SW      SW(    VT=.5 VH=.1  RON=1 ROFF=100MEG)

.end

